
CODEERZEUGUNG

Übersicht

• Verfügbare Targets im Backend: x86, ARM, PowerPC, RISC-V, … 
• Alle nicht-exotischen Prozessor-Befehlssätze ähneln sich, daher sind in 

den entsprechenden Targets ähnliche Aufgaben zu erledigen 
• Exotische Backends (C++, Verilog, …) sind natürlich auch möglich 

• The LLVM Target-Independent Code Generator 
• Algorithmen und Zwischendarstellungen architekturneutral 

implementiert 
• Targets enthalten viel Code zur Beschreibung ihrer Eigenschaften 

• TableGen-Format (*.td) 
• Targets können außerdem Hooks implementieren, wenn besondere 

Anforderungen bestehen

• Plan: Weg einer Division durch das RISC-V-
Target 

• RISC-V ??? 
• Moderner, offener/freier Befehlssatz gedacht 

für alles zwischen Embedded … HPC 
• Modular durch Befehlssatzerweiterungen 

• Sauber implementiertes LLVM-Target ohne 
Altlasten

Preface

This is version 2.2 of the document describing the RISC-V user-level architecture. The document
contains the following versions of the RISC-V ISA modules:

Base Version Frozen?
RV32I 2.0 Y
RV32E 1.9 N
RV64I 2.0 Y
RV128I 1.7 N

Extension Version Frozen?
M 2.0 Y
A 2.0 Y
F 2.0 Y
D 2.0 Y
Q 2.0 Y
L 0.0 N
C 2.0 Y
B 0.0 N
J 0.0 N
T 0.0 N
P 0.1 N
V 0.2 N
N 1.1 N

To date, no parts of the standard have been o�cially ratified by the RISC-V Foundation, but
the components labeled “frozen” above are not expected to change during the ratification process
beyond resolving ambiguities and holes in the specification.

The major changes in this version of the document include:

• The previous version of this document was released under a Creative Commons Attribution
4.0 International Licence by the original authors, and this and future versions of this document
will be released under the same licence.

• Rearranged chapters to put all extensions first in canonical order.

• Improvements to the description and commentary.

• Modified implicit hinting suggestion on JALR to support more e�cient macro-op fusion of
LUI/JALR and AUIPC/JALR pairs.

i

Zwischendarstellungen

SelectionDAGLLVM-IR Machine IR MC Layer

# Machine code for function foo: IsSSA, TracksLiveness 
Function Live Ins: $x10 in %0, $x11 in %1 

bb.0.entry: 
  liveins: $x10, $x11 
  %1:gpr = COPY $x11 
  %0:gpr = COPY $x10 
  %2:gpr = SLLI %1:gpr, 16 
  %3:gpr = SRAI killed %2:gpr, 16 
  %4:gpr = SLLI %0:gpr, 19 
  %5:gpr = SRAI killed %4:gpr, 16 
  %6:gpr = DIV killed %5:gpr, killed %3:gpr 
  %7:gpr = SLLI killed %6:gpr, 16 
  %8:gpr = SRAI killed %7:gpr, 16 
  $x10 = COPY %8:gpr 
  PseudoRET implicit $x10 

# End machine code for function foo.

 .text 
 .file "test.c" 
 .globl foo 
 .p2align 2 
 .type foo,@function 

foo: 
 addi sp, sp, -16             # encoding: [0x13,0x01,0x01,0xff] 
 sw ra, 12(sp)              # encoding: [0x23,0x26,0x11,0x00] 
 sw s0, 8(sp)               # encoding: [0x23,0x24,0x81,0x00] 
 addi s0, sp, 16              # encoding: [0x13,0x04,0x01,0x01] 
 slli a1, a1, 16              # encoding: [0x93,0x95,0x05,0x01] 
 srai a1, a1, 16              # encoding: [0x93,0xd5,0x05,0x41] 
 slli a0, a0, 19              # encoding: [0x13,0x15,0x35,0x01] 
 srai a0, a0, 16              # encoding: [0x13,0x55,0x05,0x41] 
 div a0, a0, a1              # encoding: [0x33,0x45,0xb5,0x02] 
 slli a0, a0, 16              # encoding: [0x13,0x15,0x05,0x01] 
 srai a0, a0, 16              # encoding: [0x13,0x55,0x05,0x41] 
 lw s0, 8(sp)               # encoding: [0x03,0x24,0x81,0x00] 
 lw ra, 12(sp)              # encoding: [0x83,0x20,0xc1,0x00] 
 addi sp, sp, 16              # encoding: [0x13,0x01,0x01,0x01] 
 ret                             # encoding: [0x67,0x80,0x00,0x00]

legalize-types input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

0 1
CopyFromReg

t2
i32 ch

Register %1
t3

i32

0 1
CopyFromReg

t4
i32 ch

0
truncate

t5
i16

0
truncate

t6
i16

Constant<3>
t8

i32

0 1
shl
t9

i16

0 1
sdiv
t10
i16

0
sign_extend

t11
i32

Register $x10
t12
i32

0 1 2
CopyToReg

t13
ch glue

0 1 2
RISCVISD::RET_FLAG

t14
ch

GraphRoot

LLVM-IR —> SelectionDAG

define i32 @foo(i32 %a, i32 %b) #0 { 
entry: 
  %a16 = trunc i32 %a to i16 
  %b16 = trunc i32 %b to i16 
  %shl = shl i16 %a16, 3 
  %div = sdiv i16 %shl, %b16 
  %div32 = sext i16 %div to i32 
  ret i32 %div32 
}

legalize-types input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

0 1
CopyFromReg

t2
i32 ch

Register %1
t3

i32

0 1
CopyFromReg

t4
i32 ch

0
truncate

t5
i16

0
truncate

t6
i16

Constant<3>
t8

i32

0 1
shl
t9

i16

0 1
sdiv
t10
i16

0
sign_extend

t11
i32

Register $x10
t12
i32

0 1 2
CopyToReg

t13
ch glue

0 1 2
RISCVISD::RET_FLAG

t14
ch

GraphRoot

void SelectionDAGBuilder::visitSDiv(const User &I) { 
  SDValue Op1 = getValue(I.getOperand(0)); 
  SDValue Op2 = getValue(I.getOperand(1)); 

  SDNodeFlags Flags; 
  Flags.setExact(isa<PossiblyExactOperator>(&I) && 
                 cast<PossiblyExactOperator>(&I)->isExact()); 
  setValue(&I, DAG.getNode(ISD::SDIV, getCurSDLoc(), 
        Op1.getValueType(), Op1, Op2, Flags)); 
} 

SelectionDAG

• SelectionDAG ist ein gerichteter, 
azyklischer Graph pro Basicblock 
• Knotentypen: viele (siehe rechts) 

• Target kann weitere hinzufügen 
• Kanten: Daten und „Chain“ zur 

Erhaltung der Ordnung 

• Kein Target unterstützt alle 
Operationen/Datentypen 
• Konzept des legalen/illegalen 

SelectionDAGs

… 
ADD, SUB, MUL, SDIV, UDIV, SREM, UREM, 
SMUL_LOHI, UMUL_LOHI, 
SDIVREM, UDIVREM, 
ADDC, SUBC, 
ADDE, SUBE, 
ADDCARRY, SUBCARRY, 
SADDO, UADDO, 
SSUBO, USUBO, 
SMULO, UMULO, 
FADD, FSUB, FMUL, FDIV, FREM, 
FMA, FMAD, FCOPYSIGN, 
FGETSIGN, FCANONICALIZE, 
BUILD_VECTOR, INSERT_VECTOR_ELT, 
EXTRACT_VECTOR_ELT, CONCAT_VECTORS, 
INSERT_SUBVECTOR, EXTRACT_SUBVECTOR, 
VECTOR_SHUFFLE, 
SCALAR_TO_VECTOR, 
MULHU, MULHS, 
SMIN, SMAX, UMIN, UMAX, 
AND, OR, XOR, 
ABS, 
SHL, SRA, SRL, ROTL, ROTR, 
BSWAP, CTTZ, CTLZ, CTPOP, BITREVERSE, 
CTTZ_ZERO_UNDEF, CTLZ_ZERO_UNDEF, 
SELECT, 
… 

SelectionDAG - Typ-Legalisierung

• Typ-Legalisierung stellt sicher, dass nur 
noch vom Target unterstützte 
Datentypen vorkommen 
• Hier: nur 32-bit 

• Nebenher laufen lokale Optimierungen 
auf SD (shl „vor“ sign_extend)

legalize input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

0 1
CopyFromReg

t2
i32 ch

Register %1
t3

i32

0 1
CopyFromReg

t4
i32 ch

Constant<3>
t8

i32

Register $x10
t12
i32

0 1 2
CopyToReg

t13
ch glue

0 1
sign_extend_inreg

t20
i32

ValueType:i16
t16
ch

0 1
sdiv
t19
i32

0 1 2
RISCVISD::RET_FLAG

t14
ch

0 1
shl
t15
i32

0 1
sign_extend_inreg

t17
i32

0 1
sign_extend_inreg

t18
i32

GraphRoot

// RISCVISelLowering.cpp 
addRegisterClass(XLenVT, &RISCV::GPRRegClass); 

// zum Vergleich: X86ISelLowering.cpp 
addRegisterClass(MVT::i8, &X86::GR8RegClass); 
addRegisterClass(MVT::i16, &X86::GR16RegClass); 
addRegisterClass(MVT::i32, &X86::GR32RegClass); 
if (Subtarget.is64Bit()) 
  addRegisterClass(MVT::i64, &X86::GR64RegClass); 

SelectionDAG - Legalisierung der 
Operationen

• RISC-V-Basis-ISA hat keine 
Multiplikation/Division 
• Legalisierung des SD durch 

Aufruf von Library-Funktion 
• Custom-Lowering (durch C++-

Code) möglich

isel input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

Register %1
t3

i32

Register $x10
t12
i32

0 1
CopyFromReg

t2
i32 ch

0 1
CopyFromReg

t4
i32 ch

0 1 2
CopyToReg

t13
ch glue

0 1
sra
t23
i32

Constant<16>
t21
i32

0 1
shl
t22
i32

0 1 2
RISCVISD::RET_FLAG

t14
ch

0 1 2
CopyFromReg

t34
i32 ch glue

0 1 2 3
callseq_end

t33
ch glue

TargetConstant<0>
t25
i32

0 1 2
callseq_start

t26
ch glue

0 1 2
CopyToReg

t27
ch glue

0 1
sra
t36
i32

0 1
shl
t40
i32

Register $x11
t28
i32

0 1 2 3
CopyToReg

t29
ch glue

0 1
sra
t38
i32

0 1
shl
t37
i32

TargetExternalSymbol'__divsi3'
t30
i32

RegisterMask
t31

Untyped

0 1 2 3 4 5
RISCVISD::CALL

t32
ch glue

Constant<19>
t39
i32

GraphRoot

// RISCVISelLowering.cpp 
if (!Subtarget.hasStdExtM()) { 
  setOperationAction(ISD::MUL, XLenVT, Expand); 
  setOperationAction(ISD::MULHS, XLenVT, Expand); 
  setOperationAction(ISD::MULHU, XLenVT, Expand); 
  setOperationAction(ISD::SDIV, XLenVT, Expand); 
  setOperationAction(ISD::UDIV, XLenVT, Expand); 
  setOperationAction(ISD::SREM, XLenVT, Expand); 
  setOperationAction(ISD::UREM, XLenVT, Expand); 
} 

SelectionDAG - Legalisierung der 
Operationen

• RISC-V-ISA-Erweiterung „m“ stellt 
Multiplikation/Division bereit 

• sign_extend_inreg wurde zu shl/sra-
Paar

isel input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

Register %1
t3

i32

Register $x10
t12
i32

0 1
CopyFromReg

t2
i32 ch

0 1
CopyFromReg

t4
i32 ch

0 1
sdiv
t19
i32

0 1
sra
t25
i32

0 1
sra
t27
i32

Constant<16>
t21
i32

0 1
shl
t29
i32

0 1
shl
t26
i32

0 1 2
CopyToReg

t13
ch glue

0 1
sra
t23
i32

0 1
shl
t22
i32

0 1 2
RISCVISD::RET_FLAG

t14
ch

Constant<19>
t28
i32

GraphRoot

// RISCVInstrInfoM.td 

let Predicates = [HasStdExtM] in { 
def MUL     : ALU_rr<0b0000001, 0b000, "mul">; 
def MULH    : ALU_rr<0b0000001, 0b001, "mulh">; 
def MULHSU  : ALU_rr<0b0000001, 0b010, "mulhsu">; 
def MULHU   : ALU_rr<0b0000001, 0b011, "mulhu">; 
def DIV     : ALU_rr<0b0000001, 0b100, "div">; 
def DIVU    : ALU_rr<0b0000001, 0b101, "divu">; 
def REM     : ALU_rr<0b0000001, 0b110, "rem">; 
def REMU    : ALU_rr<0b0000001, 0b111, "remu">; 
} // Predicates = [HasStdExtM] 

TableGen

• Targets müssen viele domänenspezifische Informationen für den 
Codegenerator zur Verfügung stellen 
• Instruktionen, Registersatz, … 

• TableGen dient zur Beschreibung solcher Records 
• Ziel: Möglichst viele Gemeinsamkeiten ausfaktorisieren, um 

Beschreibung übersichtlich zu halten 
• Flow: TableGen-Tool liest .td-Datei, instanziiert Records, und gibt sie an 

domänenspezifisches Backend weiter 
• … häufig werden C++-Definitionen erzeugt, die anderswo inkludiert 

werden

TableGen: Beispiel

class klass<string Arg1, string Arg2> { 
  string message =  
    "hello" # Arg1 # " -> " # Arg2; 
} 
class klass2<int n> { 
  int N = n; 
} 

def foo : klass<"world", „kthxbai">; 

let message = "overriding..." in { 
  def bar : klass<"doesnt", "matter">; 
} 
def baz :  
  klass<"number", "see below“>, 
  klass2<42>; 

def blub { 
  string here = "you go"; 
}

------------- Classes ----------------- 
class klass<string klass:Arg1 = ?, string klass:Arg2 = ?> { 
  string message = !strconcat("hello", !strconcat(klass:Arg1, !strconcat(" -> ", 
klass:Arg2))); 
} 
class klass2<int klass2:n = ?> { 
  int N = klass2:n; 
} 
------------- Defs ----------------- 
def bar { // klass 
  string message = "overriding..."; 
} 
def baz { // klass klass2 
  string message = "hellonumber -> see below"; 
  int N = 42; 
} 
def blub { 
  string here = "you go"; 
} 
def foo { // klass 
  string message = "helloworld -> kthxbai"; 
}

SelectionDAG: Instruction Selection

• Mapping von SDNodes auf Opcodes des 
Targets 

• Pattern Matching wird automatisch aus 
TableGen-Beschreibung des 
Instruktionssatzes generiert

!12

scheduler input for foo:entry

EntryToken
t0
ch

Register %0
t1

i32

Register %1
t3

i32

Register $x10
t12
i32

0 1
CopyFromReg

t2
i32 ch

0 1
CopyFromReg

t4
i32 ch

0 1
SLLI
t29
i32

TargetConstant<19>
t31
i32

0 1
SLLI
t26
i32

TargetConstant<16>
t30
i32

0 1
SRAI

t25
i32

0 1
SRAI

t27
i32

0 1
DIV
t19
i32

0 1
SLLI
t22
i32

0 1
SRAI

t23
i32

0 1 2
CopyToReg

t13
ch glue

0 1 2
PseudoRET

t14
ch

GraphRoot

def SLLI : Shift_ri<0, 0b001, "slli">; 
def SRAI : Shift_ri<1, 0b101, "srai">; 

def DIV     : ALU_rr<0b0000001, 0b100, "div">; 

class ALU_rr<bits<7> funct7, bits<3> funct3, string opcodestr> 
    : RVInstR<funct7, funct3, OPC_OP, 

(outs GPR:$rd), (ins GPR:$rs1, GPR:$rs2), 
              opcodestr, "$rd, $rs1, $rs2“>; 
class Shift_ri<bit arithshift, bits<3> funct3, string opcodestr> 
    : RVInstIShift<arithshift, funct3, OPC_OP_IMM, (outs GPR:$rd), 
                   (ins GPR:$rs1, uimmlog2xlen:$shamt), opcodestr, 
                   "$rd, $rs1, $shamt">;

SelectionDAG —> Machine IR

• Machine IR besteht wieder aus 
(Machine-)Functions, 
BasicBlock, Instructions 

• Scheduler berechnet Ordnung 
der MachineInstructions 

• Code verwendet virtuelle 
Register, ist weiterhin in SSA-
Form

!13

# Machine code for function foo: IsSSA, 
# TracksLiveness 
Function Live Ins: $x10 in %0, $x11 in %1 

bb.0.entry: 
  liveins: $x10, $x11 
  %1:gpr = COPY $x11 
  %0:gpr = COPY $x10 
  %2:gpr = SLLI %1:gpr, 16 
  %3:gpr = SRAI killed %2:gpr, 16 
  %4:gpr = SLLI %0:gpr, 19 
  %5:gpr = SRAI killed %4:gpr, 16 
  %6:gpr = DIV killed %5:gpr, killed %3:gpr 
  %7:gpr = SLLI killed %6:gpr, 16 
  %8:gpr = SRAI killed %7:gpr, 16 
  $x10 = COPY %8:gpr 
  PseudoRET implicit $x10 

# End machine code for function foo.

Machine IR 

• Machine IR wird nochmal ausgiebig optimiert: 
• Routinenprotokoll, Registerallokation, …

!14

Expand ISel Pseudo-instructions: 
Early Tail Duplication: 
Optimize machine instruction PHIs: 
Merge disjoint stack slots: 
Local Stack Slot Allocation: 
Remove dead machine instructions: 
Early Machine Loop Invariant Code Motion: 
Machine Common Subexpression Elimination: 
Machine code sinking: 
Peephole Optimizations: 
Remove dead machine instructions: 
Detect Dead Lanes: 
Process Implicit Definitions: 
Live Variable Analysis: 
Machine Natural Loop Construction: 
Eliminate PHI nodes for register allocation: 
Two-Address instruction pass: 
Simple Register Coalescing: 
Rename Disconnected Subregister Components: 
Machine Instruction Scheduler: 
Greedy Register Allocator: 
Virtual Register Rewriter: 
Stack Slot Coloring: 
Machine Copy Propagation Pass: 
Machine Loop Invariant Code Motion: 
PostRA Machine Sink: 
Shrink Wrapping analysis: 
Prologue/Epilogue Insertion & Frame Finalization: 
Control Flow Optimizer: 
Tail Duplication: 
Machine Copy Propagation Pass: 
Post-RA pseudo instruction expansion pass: 
Post RA top-down list latency scheduler: 
Analyze Machine Code For Garbage Collection: 
Branch Probability Basic Block Placement: 
Branch relaxation pass: 
Contiguously Lay Out Funclets: 
StackMap Liveness Analysis: 
Live DEBUG_VALUE analysis: 
Insert fentry calls: 
Insert XRay ops: 
Implement the 'patchable-function' attribute:

# Machine code for function foo: NoPHIs, TracksLiveness, NoVRegs 
Frame Objects: 
  fi#0: size=4, align=4, at location [SP-4] 
  fi#1: size=4, align=4, at location [SP-8] 
Function Live Ins: $x10, $x11 

bb.0.entry: 
  liveins: $x10, $x11 
  $x2 = frame-setup ADDI $x2, -16 
  SW killed $x1, $x2, 12 
  SW killed $x8, $x2, 8 
  $x8 = frame-setup ADDI $x2, 16 
  renamable $x11 = SLLI killed renamable $x11, 16 
  renamable $x11 = SRAI killed renamable $x11, 16 
  renamable $x10 = SLLI killed renamable $x10, 19 
  renamable $x10 = SRAI killed renamable $x10, 16 
  renamable $x10 = DIV killed renamable $x10, killed renamable $x11 
  renamable $x10 = SLLI killed renamable $x10, 16 
  renamable $x10 = SRAI killed renamable $x10, 16 
  $x8 = LW $x2, 8 
  $x1 = LW $x2, 12 
  $x2 = frame-destroy ADDI $x2, 16 
  PseudoRET implicit $x10 

Machine IR —> MC Layer

• Abstraktionsebene: Sections, Labels, Instructions, … 
• Ausgabe entweder als Assemblercode oder binär

!15

 .text 
 .file "test.c" 
 .globl foo 
 .p2align 2 
 .type foo,@function 
foo: 
 addi sp, sp, -16             # encoding: [0x13,0x01,0x01,0xff] 
 sw ra, 12(sp)              # encoding: [0x23,0x26,0x11,0x00] 
 sw s0, 8(sp)               # encoding: [0x23,0x24,0x81,0x00] 
 addi s0, sp, 16              # encoding: [0x13,0x04,0x01,0x01] 
 slli a1, a1, 16              # encoding: [0x93,0x95,0x05,0x01] 
 srai a1, a1, 16              # encoding: [0x93,0xd5,0x05,0x41] 
 slli a0, a0, 19              # encoding: [0x13,0x15,0x35,0x01] 
 srai a0, a0, 16              # encoding: [0x13,0x55,0x05,0x41] 
 div a0, a0, a1              # encoding: [0x33,0x45,0xb5,0x02] 
 slli a0, a0, 16              # encoding: [0x13,0x15,0x05,0x01] 
 srai a0, a0, 16              # encoding: [0x13,0x55,0x05,0x41] 
 lw s0, 8(sp)               # encoding: [0x03,0x24,0x81,0x00] 
 lw ra, 12(sp)              # encoding: [0x83,0x20,0xc1,0x00] 
 addi sp, sp, 16              # encoding: [0x13,0x01,0x01,0x01] 
 ret                              # encoding: [0x67,0x80,0x00,0x00]

Machine IR —> MC Layer

• Details des Instruktionsformats sind als TableGen-Records kodiert 
• Assembler-Darstellung ebenfalls

!16

def DIV     : ALU_rr<0b0000001, 0b100, "div">; 

class ALU_rr<bits<7> funct7, bits<3> funct3, string opcodestr> 
    : RVInstR<funct7, funct3, OPC_OP, 

(outs GPR:$rd), (ins GPR:$rs1, GPR:$rs2), 
              opcodestr, "$rd, $rs1, $rs2“>; 

class RVInstR<bits<7> funct7, bits<3> funct3, RISCVOpcode opcode, dag outs, 
              dag ins, string opcodestr, string argstr> 
    : RVInst<outs, ins, opcodestr, argstr, [], InstFormatR> { 
  bits<5> rs2; 
  bits<5> rs1; 
  bits<5> rd; 

  let Inst{31-25} = funct7; 
  let Inst{24-20} = rs2; 
  let Inst{19-15} = rs1; 
  let Inst{14-12} = funct3; 
  let Inst{11-7} = rd; 
  let Opcode = opcode.Value; 
} 

36 Volume I: RISC-V User-Level ISA V2.2

obtain the upper 32 bits of the 64-bit product, but signed arguments must be proper 32-bit signed
values, whereas unsigned arguments must have their upper 32 bits clear.

6.2 Division Operations

31 25 24 20 19 15 14 12 11 7 6 0

funct7 rs2 rs1 funct3 rd opcode
7 5 5 3 5 7

MULDIV divisor dividend DIV[U]/REM[U] dest OP
MULDIV divisor dividend DIV[U]W/REM[U]W dest OP-32

DIV and DIVU perform signed and unsigned integer division of XLEN bits by XLEN bits. REM
and REMU provide the remainder of the corresponding division operation. If both the quotient
and remainder are required from the same division, the recommended code sequence is: DIV[U]
rdq, rs1, rs2; REM[U] rdr, rs1, rs2 (rdq cannot be the same as rs1 or rs2). Microarchitectures can
then fuse these into a single divide operation instead of performing two separate divides.

The semantics for division by zero and division overflow are summarized in Table 6.1. The quotient
of division by zero has all bits set, i.e. 2XLEN � 1 for unsigned division or �1 for signed division.
The remainder of division by zero equals the dividend. Signed division overflow occurs only when
the most-negative integer, �2XLEN�1, is divided by �1. The quotient of signed division overflow
is equal to the dividend, and the remainder is zero. Unsigned division overflow cannot occur.

Condition Dividend Divisor DIVU REMU DIV REM
Division by zero x 0 2XLEN � 1 x �1 x
Overflow (signed only) �2XLEN�1 �1 – – �2XLEN�1 0

Table 6.1: Semantics for division by zero and division overflow.

We considered raising exceptions on integer divide by zero, with these exceptions causing a trap in
most execution environments. However, this would be the only arithmetic trap in the standard
ISA (floating-point exceptions set flags and write default values, but do not cause traps) and
would require language implementors to interact with the execution environment’s trap handlers
for this case. Further, where language standards mandate that a divide-by-zero exception must
cause an immediate control flow change, only a single branch instruction needs to be added to
each divide operation, and this branch instruction can be inserted after the divide and should
normally be very predictably not taken, adding little runtime overhead.

The value of all bits set is returned for both unsigned and signed divide by zero to simplify
the divider circuitry. The value of all 1s is both the natural value to return for unsigned divide,
representing the largest unsigned number, and also the natural result for simple unsigned divider
implementations. Signed division is often implemented using an unsigned division circuit and
specifying the same overflow result simplifies the hardware.

DIVW and DIVUW instructions are only valid for RV64, and divide the lower 32 bits of rs1 by
the lower 32 bits of rs2, treating them as signed and unsigned integers respectively, placing the
32-bit quotient in rd, sign-extended to 64 bits. REMW and REMUW instructions are only valid
for RV64, and provide the corresponding signed and unsigned remainder operations respectively.

Quellen / weiterführende Infos

• https://www.llvm.org/docs/CodeGenerator.html 
• https://www.llvm.org/docs/WritingAnLLVMBackend.html 
• https://riscv.org/specifications/ 
• https://eli.thegreenplace.net/2012/11/24/life-of-an-instruction-in-llvm/ 
• https://content.riscv.org/wp-content/uploads/2018/05/

asb_llvm_2018_barcelona_riscv_workshop_slides.pdf 
• https://git.llvm.org/git/llvm.git

